Geometry of Surfaces of Revolution

Last Revision: 16 March 2017

Justin M. Ryan

Email: ryan@math.wichita.edu

Let f be a positive, smooth function on an open interval (a,b). Then f defines a smooth surface (2-dimensional manifold) S by rotating the graph of f about the x-axis. Moreover, S is covered by two charts U_1=(a,b) x (-π,π) and U_2=(a,b) x (0,2π) with parametrizations ψ_i(t,θ) = (t,f(t)cosθ,f(t)sinθ). The chart maps are defined by ϕ_1(x,y,z) = $\left(x,\ \sin^{-1}\left(\frac{y}{\sqrt{y^2+z^2}}\right)\right)$ and ϕ_2(x,y,z) = $\left(x,\ \cos^{-1}\left(\frac{y}{\sqrt{y^2+z^2}}\right)\right)$.

In this notebook, I compute the induced metric on the surface inherited from the Euclidean metric on R^3, the Christoffel symbols (Levi-Civita connection coefficients), and some other differential geometric properties. I've tried to design the code so that the

Suggestions for improvements are welcomed.

First we define the function and the open interval (a,b):

```
f[x_] := F[x];
(*
a=π/2;
b=7π/2;
*)
```

The parametrization and charts are given by:

```
param[t_, θ_] := {t, f[t] Sin[θ], f[t] Cos[θ]};
ϕ1[x_, y_, z_] := {x, ArcSin[y/Sqrt[y^2 + z^2]]};
ϕ2[x_, y_, z_] := {x, ArcCos[y/Sqrt[y^2 + z^2]]};
```

Next we define the surface S:

```
(*
S=ParametricPlot3D[param[t,θ],{t,a,b},{θ,0,2π},ImageSize→350]
*)
```

The following code defines the induced metric *via* the parametrization map.

```
metric[x_][u_,v_]:=
Module[{xu,xv,U,V,si},
si=Simplify[#]/.{U→u,V→v}&;
xu=D[x[U,V],U];
xv=D[x[U,V],V];
ee=xu.xu//si;
ff=xu.xv//si;
gg=xv.xv//si;
{{ee,ff},{ff,gg}}
]
```

```
g = metric[param][t, θ];
g // MatrixForm
```

$$\begin{pmatrix} 1 + F'[t]^2 & 0 \\ 0 & F[t]^2 \end{pmatrix}$$

We now compute the Christoffel symbols $\Gamma_{ij}{}^{k}$=Γ(i,j,k) for the Levi-Civita connection on S.

```
Γ[i_,j_,k_]:=
Module[{vv,comp},
vv={t,θ};
comp=Simplify[1/2 Sum[Inverse[metric[param][t,θ]][[k,m]] (D[metric[param][t,θ][[m,i]],vv[
 +D[metric[param][t,θ][[m,j]],vv[[i]]]
 -D[metric[param][t,θ][[i,j]],vv[[m]]]),{m,1,2}]];
Clear[vv];
comp]
```

The only non-zero symbols will be Γ(1,1,1) , Γ(2,2,1), Γ(1,2,2), and Γ(2,1,2).

```
Γ[1, 1, 1]
```

$$\frac{F'[t]\, F''[t]}{1 + F'[t]^2}$$

```
Γ[2, 2, 1]
```

$$-\frac{F[t]\, F'[t]}{1 + F'[t]^2}$$

```
Γ[1, 2, 2]
```

$$\frac{F'[t]}{F[t]}$$

```
Γ[2, 1, 2]
```

$$\frac{F'[t]}{F[t]}$$

The Christoffel symbols can be collected into an array for ease of viewing, if nothing else.

```
CS=Array[Γ,{2,2,2}];
```

```
CS // MatrixForm
```

$$\begin{pmatrix} \begin{pmatrix} \frac{F'[t]\,F''[t]}{1+F'[t]^2} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{F'[t]}{F[t]} \end{pmatrix} \\ \begin{pmatrix} 0 \\ \frac{F'[t]}{F[t]} \end{pmatrix} & \begin{pmatrix} -\frac{F[t]\,F'[t]}{1+F'[t]^2} \\ 0 \end{pmatrix} \end{pmatrix}$$

The Riemann curvature is given by R(x,y)z := $\nabla_x \nabla_y z - \nabla_y \nabla_x z - \nabla_{[x,y]} z$. The curvature components can be computed in local coordinates in terms of the Christoffel symbols.

```
Riem[i_,j_,k_,l_]:=
Module[{vv,curv,s1,s2},
vv={t,θ};
s1=Sum[Γ[l,m,i]Γ[k,j,m],{m,1,2}];
s2=Sum[Γ[k,m,i]Γ[l,j,m],{m,1,2}];
curv=Simplify[
D[Γ[k,j,i],vv[[l]]] - D[Γ[l,j,i],vv[[k]]] +s1-s2];
Clear[s1,s2,vv];
curv]
```

Again, it will be useful to collect all of the curvature components into a stack of matrices.

```
R = Array[Riem,{2,2,2,2}];
```

```
R // MatrixForm
```

$$\begin{pmatrix} \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} & \begin{pmatrix} 0 & \frac{F[t]\,F''[t]}{(1+F'[t]^2)^2} \\ -\frac{F[t]\,F''[t]}{(1+F'[t]^2)^2} & 0 \end{pmatrix} \\ \begin{pmatrix} 0 & -\frac{F''[t]}{F[t]+F[t]\,F'[t]^2} \\ \frac{F''[t]}{F[t]+F[t]\,F'[t]^2} & 0 \end{pmatrix} & \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} \end{pmatrix}$$

Ricci curvature is the contraction of the Riemann curvature. It is obtained by taking the traces Ric(x,y) = tr($\xi \to$ R(x,ξ)y).

```
Ric=TensorContract[R,{1,3}];
```

```
Ric // MatrixForm
```

$$\begin{pmatrix} \frac{F''[t]}{F[t]+F[t]\,F'[t]^2} & 0 \\ 0 & \frac{F[t]\,F''[t]}{(1+F'[t]^2)^2} \end{pmatrix}$$

The Ricci operator is the linear operator metrically equivalent to Ric. It is given by

```
Rc=Inverse[g].Ric;
```

```
Rc // MatrixForm
```

$$\begin{pmatrix} \frac{F[t]^2 F''[t]}{(F[t]+F[t] F'[t]^2)(F[t]^2+F[t]^2 F'[t]^2)} & 0 \\ 0 & \frac{F[t] F''[t]}{(1+F'[t]^2)(F[t]^2+F[t]^2 F'[t]^2)} \end{pmatrix}$$

Scalar curvature is the contraction (or trace) of the Ricci operator.

```
sc = Simplify[Tr[Rc]];
```

```
sc
```

$$\frac{2 F''[t]}{F[t] \left(1+F'[t]^2\right)^2}$$

On a surface *S*, sectional curvature is a function $\kappa : S \to R$. (There is only one tangent plane at each point.)

```
Q=Det[g];
κ=Simplify[Riem[1,2,1,2]/Q];
```

```
κ
```

$$\frac{F''[t]}{F[t] \left(1+F'[t]^2\right)^3}$$

Geodesics of (*S*, *g*) are curves γ: R → *S* that satisfy $\nabla_{\gamma'} \gamma' = 0$, where γ' is the velocity vector field along the image of γ.

```
γ[t_]:={x[t],y[t]};
```

```
equation1:=D[D[γ[t][[1]],t],t]+Sum[Sum[Γ[i,j,1]D[γ[t][[i]],t]D[γ[t][[j]],t],{i,1,2}],{
equation2:=D[D[γ[t][[2]],t],t]+Sum[Sum[Γ[i,j,2]D[γ[t][[i]],t]D[γ[t][[j]],t],{i,1,2}],{
```

```
equation1
```

$$-\frac{F[t] F'[t] y'[t]^2}{1+F'[t]^2}+\frac{F'[t] x'[t]^2 F''[t]}{1+F'[t]^2}+x''[t]==0$$

```
equation2
```

$$\frac{2 F'[t] x'[t] y'[t]}{F[t]}+y''[t]==0$$

```
initcond = {x[0] == 0, y[0] == 0};
DSolve[{equation1, equation2}, {x[t], y[t]}, t]
```

$$\text{DSolve}\left[\left\{-\frac{F[t] F'[t] y'[t]^2}{1+F'[t]^2}+\frac{F'[t] x'[t]^2 F''[t]}{1+F'[t]^2}+x''[t]==0,\ \frac{2 F'[t] x'[t] y'[t]}{F[t]}+y''[t]==0\right\},\ \{x[t], y[t]\}, t\right]$$